Illustrate music beats You know—they look like this:


source

The goal is to draw a music beats illustration like the following:

=     =      =
=  =  =      =          =
== = ==   =  ==     = ====
== ====  == ===   = = =======
======== == ====  = ========= =
=================================


The rules are:


The width of the illustration is 33 symbols, but if you need—any trailing spaces exceeding this width are allowed.
Each column is made of equals signs (=).
Each column has a random height (the height of the next column shouldn’t depend in any way on the height of the previous column), varying from 1 to 6. It’s also fine if it’s at least possible to get some input with no strict math probability (i.e. some inputs could appear more rarely than others).
A column can’t float above the bottom and have gaps in it.
Since every column has the minimal height of 1, the last row can’t have any gaps either—it always consists of 33 equals signs.
Since it’s possible to have no columns with the height of 6 (it’s all random after all): in this case you don’t need to have a top line made of spaces. Applies to any edge cases of this nature: if suddenly your code provided no columns with the height greater than 1, you don’t need to have additional lines made of spaces above the bottom line.
You don’t take any input.
Dyalog APL, 14 bytes
⊖⍉↑'='⍴¨⍨?33⍴6

Explanation
33⍴6 33 repetitions of 6
? random integer in range [1,n] for each of the 33 6s
'='⍴¨⍨ equality symbol repeated each of those number of times
↑ convert list of lists to table of rows
⍉ transpose rows into columns, columns into rows
⊖ flip upside down
Example runs
Input is indented six spaces:
      ⊖⍉↑'='⍴¨⍨?33⍴6
=          ==        =      ==   
= =    =   ==      ====     ==   
= = = ===  ==  === ==== =  ===  =
= = ===== ==== === ==== = ====  =
=== ============== ==== ====== ==
=================================
      ⊖⍉↑'='⍴¨⍨?33⍴6
         =  =  =    =    =       
  =      =  =  ==  == == =  =    
 === == ==  =  === =======  =    
==== == ==  ====== ==========   =
==== ============= ========== = =
=================================
      ⊖⍉↑'='⍴¨⍨?33⍴6
             =    =   =  =       
         =   =    =   = == =     
=      = =   =    ==  = ==== === 
= = =  = =   =    ==  = ==== ====
=====  = == ==  ============ ====
=================================